Electronic Supplementary Material (ESI) for Soft Matter.

In[1]:=
```
(*** Input the expressions by Goldman, Cox, and Brenner ***)
```

In[2]:=
```
(* Shifted the lubrication curves to match the numerical data by O'Neill for ξ = δ/a to order O(δ/a) *)

Ftx[ξ_] := (8/15) * Log[ξ] - 0.9588;
Tty[ξ_] := - (1/10) * Log[ξ] - 0.1895;
Frx[ξ_] := - (2/15) * Log[ξ] - 0.2526;
Try[ξ_] := (2/5) * Log[ξ] - 0.3817;
```

In[6]:=
```
(* Plots for the translating sphere case, these match the lines in the Goldman paper *)

LogLinearPlot[-Ftx[d], {d, 0.001, 10}, PlotStyle → {Thick, Red}, TicksStyle → Directive[Black, Thick, 30],
 AxesLabel → {Style["δ/a", Black, Thick, 30], Style["-Fx/(6πμaU)", Black, Thick, 30]}, ImageSize → 750]
LogLinearPlot[Tty[d], {d, 0.001, 10}, PlotStyle → {Thick, Red}, TicksStyle → Directive[Black, Thick, 30],
 AxesLabel → {Style["δ/a", Black, Thick, 30], Style["Ty/(8πμa²U)", Black, Thick, 30]}, ImageSize → 750]
```

Out[6]=



Out[7]=



In[8]:=

```
(* Plots for the rotating sphere case, these too match the lines in the paper *)

LogLinearPlot[Frx[d], {d, 0.001, 10}, PlotStyle → {Thick, Red}, TicksStyle → Directive[Black, Thick, 30],
 AxesLabel → {Style["δ/a", Black, Thick, 30], Style["Fx/(6πμa²Ω)", Black, Thick, 30]}, ImageSize → 750]
LogLinearPlot[-Try[d], {d, 0.00001, 10}, PlotStyle → {Thick, Red}, TicksStyle → Directive[Black, Thick, 30],
 AxesLabel → {Style["δ/a", Black, Thick, 30], Style["-Ty/(8πμa³Ω)", Black, Thick, 30]}, ImageSize → 750]
```

Out[8]=



Out[9]=



In[10]:=

```
(*** Extract the velocity and angular velocity for the applied torque ***)
```

In[11]:=
```
(* Set of equations for a sphere both translating and rotating *)

fx = 6 * π * μ * a * (U * ftx + a * Ω * frx)
ty = 8 * π * μ * a^2 * (U * tty + a * Ω * try)
```

Out[11]= $6\, a\, \pi\, \mu\, (\text{ftx}\, U + a\, \text{frx}\, \Omega)$

Out[12]= $8\, a^2\, \pi\, \mu\, (\text{tty}\, U + a\, \text{try}\, \Omega)$

In[13]:=
```
(* Solve speed/rotation ratio for no external force (fx==0) and extract the velocity *)

solU = Solve[fx == 0, Ω][[1]]
rhs = ty /. solU // Simplify;
vel = Solve[Ty == rhs, U][[1, 1, 2]]
```

Out[13]= $\left\{\Omega \to -\frac{\text{ftx}\, U}{a\, \text{frx}}\right\}$

Out[15]= $-\frac{\text{frx}\, \text{Ty}}{8\, a^2\, \pi\, (\text{ftx}\, \text{try} - \text{frx}\, \text{tty})\, \mu}$

In[16]:=
```
(* Now plug the Goldman expressions in and obtain the velocity and angular velocity in terms of the gap size *)

subs = {frx → Frx[δ / a], ftx → Ftx[δ / a], try → Try[δ / a], tty → Tty[δ / a]};
VV = vel /. subs
ΩΩ = solU[[1, 2]] /. {U → vel} /. subs
```

Out[17]= $-\frac{\text{Ty}\left(-0.2526 - \frac{2}{15}\text{Log}\left[\frac{\delta}{a}\right]\right)}{8\, a^2\, \pi\, \mu\left(-\left(-0.2526 - \frac{2}{15}\text{Log}\left[\frac{\delta}{a}\right]\right)\left(-0.1895 - \frac{1}{10}\text{Log}\left[\frac{\delta}{a}\right]\right) + \left(-0.3817 + \frac{2}{5}\text{Log}\left[\frac{\delta}{a}\right]\right)\left(-0.9588 + \frac{8}{15}\text{Log}\left[\frac{\delta}{a}\right]\right)\right)}$

Out[18]= $\frac{\text{Ty}\left(-0.9588 + \frac{8}{15}\text{Log}\left[\frac{\delta}{a}\right]\right)}{8\, a^3\, \pi\, \mu\left(-\left(-0.2526 - \frac{2}{15}\text{Log}\left[\frac{\delta}{a}\right]\right)\left(-0.1895 - \frac{1}{10}\text{Log}\left[\frac{\delta}{a}\right]\right) + \left(-0.3817 + \frac{2}{5}\text{Log}\left[\frac{\delta}{a}\right]\right)\left(-0.9588 + \frac{8}{15}\text{Log}\left[\frac{\delta}{a}\right]\right)\right)}$

In[19]:=
```
(* Determine the relevant expression for the reduced velocity and study the limiting behavior *)

frac = VV / (ΩΩ * a);
frac // Simplify

Limit[frac, δ → 0]
```

Out[20]= $\frac{0.25\left(1.8945 + \text{Log}\left[\frac{\delta}{a}\right]\right)}{-1.79775 + \text{Log}\left[\frac{\delta}{a}\right]}$

Out[21]= 0.25

In[22]:=
```
(* Lastly plot the reduced velocity as a function of swimmer-wall separation *)

pfrac = frac /. {δ → x * a} // Simplify;
LogLinearPlot[pfrac, {x, 10^(-10), 0.1}, PlotStyle → {Thick, Red}, TicksStyle → Directive[Black, Thick, 30],
 AxesLabel → {Style["δ/a", Black, Thick, 30], Style["V/(Ωa)", Black, Thick, 30]}, ImageSize → 750]
```

Out[23]=